
Subject: Re: Help: Countlines on Win 95 version of IDL
Posted by [mgs](#) on Thu, 18 Dec 1997 08:00:00 GMT
[View Forum Message](#) <> [Reply to Message](#)

In article <34981500.572F0480@fz-juelich.de>, "R. Bauer"
<r.bauer@fz-juelich.de> wrote:

> PDW wrote:

>

>> I am converting programs from Unix version of IDL to Windows 95, and have a
>> problem with routines using SPAWN and "wc -l" to get the number of lines in
>> a file in order to dimension an array. How can this be done on the PC
>> version?

Here's another "Fileline". Same purpose and name, different author. Works
cross-platform (UNIX, Mac and PC) and shouldn't gag on huge files.

```
#####  
; File Name: %M%  
; Version: %I%  
; Author: Mike Schienle  
; Orig Date: 97-03-09  
; Delta Date: %G% @ %U%  
#####  
; Purpose:  
; History:  
#####  
; %W%  
#####
```

```
FUNCTION FileLine, file, LINETERM=lineTerm  
; open the file and check for errors  
OpenR, lun, file, /Get_Lun, Error=err, BufSize=2L^20  
IF NOT (err) THEN BEGIN  
; if line terminator is not specified  
; assume the file is native to the OS  
; UNIX uses line feed  
; Mac uses carriage return  
; PC uses both  
IF (N_Elements(lineTerm) EQ 0L) THEN $  
IF (StrLowerCase(!Version.OS_Family) EQ 'unix') THEN $  
lineTerm = 10b $  
ELSE $  
lineTerm = 13b  
  
; use FStat function to get access to file size  
fInfo = FStat(lun)
```

```

; set an arbitrary size threshold
IThresh = 2L^20

; compare the file size to the threshold
IF (fInfo.Size LT IThresh) THEN BEGIN
    ; allocate array to read entire file into memory
    aData = BytArr(fInfo.Size, /NoZero)
    ReadU, lun, aData

    ; find the number of characters matching line terminator
    ; Provided in ICount. Not interested in positions.
    aPos = Where(aData EQ lineTerm, ICount)
ENDIF ELSE BEGIN
    ; file arbitrarily too large to read into a single array
    ; determine the number of loops to be processed
    ILoops = fInfo.Size / IThresh

    ; associate a variable for easy access
    mVar = Assoc(lun, BytArr(IThresh, /Nozero))

    ; init a var to hold the terminator count
    ICount = 0L

    ; loop through the data file one threshold chunk at a time
    FOR i = 0L, (ILoops - 1) DO BEGIN
        ; find the number of characters matching line terminator
        ; Provided in IPos. Not interested in positions.
        aPos = Where(mVar(i) EQ lineTerm, IPos)

        ; running sum of terminator counts
        ICount = ICount + IPos
    ENDFOR

    ; get amount of file remainder
    IModSize = (fInfo.Size MOD IThresh)

    ; read remainder of file if necessary
    IF (IModSize NE 0) THEN BEGIN
        ; create an array to read in remaining data
        aData = BytArr(IModSize, /NoZero)

        ; put the data in the file
        ReadU, lun, aData

        ; find the number of characters matching line terminator
        ; Provided in IPos. Not interested in positions.
        aPos = Where(aData EQ lineTerm, IPos)
    
```

```
        ; running sum of terminator counts
        ICount = ICount + IPos
    ENDIF
ENDELSE

    ; free and close the lun
    Free_Lun, lun
ENDIF ELSE BEGIN
    ; an error occurred
    ICount = -1
ENDELSE

; return the ICount to the calling program
Return, ICount
END
```

--

Mike Schienle
mgs@sd.cybernex.net

Interactive Visuals
<http://ww2.sd.cybernex.net/~mgs/>
